Problem:
Score : 400 points

Problem Statement

A university student, Takahashi, has to take N examinations and pass all of them.
Currently, his readiness for the i-th examination is A_{i}, and according to his investigation, it is known that he needs readiness of at least B_{i} in order to pass the i-th examination.

Takahashi thinks that he may not be able to pass all the examinations, and he has decided to ask a magician, Aoki, to change the readiness for as few examinations as possible so that he can pass all of them, while not changing the total readiness.

For Takahashi, find the minimum possible number of indices i such that A_i and C_i are different, for a sequence C_1, C_2, ..., C_{N} that satisfies the following conditions:

The sum of the sequence A_1, A_2, ..., A_{N} and the sum of the sequence C_1, C_2, ..., C_{N} are equal.

For every i, B_i \leq C_i holds.

If such a sequence C_1, C_2, ..., C_{N} cannot be constructed, print -1.

Constraints

1 \leq N \leq 10^5

1 \leq A_i \leq 10^9

1 \leq B_i \leq 10^9

A_i and B_i are integers.

Input

Input is given from Standard Input in the following format:

N
A_1 A_2 ... A_{N}
B_1 B_2 ... B_{N}

Output

Print the minimum possible number of indices i such that A_i and C_i are different, for a sequence C_1, C_2, ..., C_{N} that satisfies the conditions.
If such a sequence C_1, C_2, ..., C_{N} cannot be constructed, print -1.

Sample Input 1

3
2 3 5
3 4 1

Sample Output 1

3

(A_1, A_2, A_3) = (2, 3, 5) and (B_1, B_2, B_3) = (3, 4, 1). If nothing is done, he cannot pass the first and second exams.
The minimum possible number of indices i such that A_i and C_i are different, 3, is achieved when:

(C_1, C_2, C_3) = (3, 5, 2)

Sample Input 2

3
2 3 3
2 2 1

Sample Output 2

0

In this case, he has to do nothing in order to pass all the exams.

Sample Input 3

3
17 7 1
25 6 14

Sample Output 3

-1

In this case, no matter what is done, he cannot pass all the exams.

Sample Input 4

12
757232153 372327760 440075441 195848680 354974235 458054863 463477172 740174259 615762794 632963102 529866931 64991604
74164189 98239366 465611891 362739947 147060907 118867039 63189252 78303147 501410831 110823640 122948912 572905212

Sample Output 4

5

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

program exam_and_wizard
  implicit none
  integer :: n, i, num
  integer(8) :: a(100000), b(100000), c(100000), tot
  a = 0_8
  b = 0_8
  c = 0_8
  read(*,*) n
  read(*,*) a(1:n)
  read(*,*) b(1:n)
  c(1:n) = a(1:n) - b(1:n)
  num = 0
  tot = 0_8
  do i = 1, n
    if (c(i).lt.0) then
      num = num + 1
      tot = tot - c(i)
    end if
  end do
  if (num.eq.0) then
    write(*,'(i0)') 0
    stop
  end if
  call merge_sort(n,c(1:n))
  do i = n, 1, -1
    if (c(i).lt.0) exit
    tot = tot - c(i)
    num = num + 1
    if (tot.le.0_8) then
      write(*,'(i0)') num
      stop
    end if
  end do
  write(*,'(i0)') -1
  stop
contains
  subroutine merge_sort(n,c)
    implicit none
    integer, intent(in) :: n
    integer(8), intent(inout) :: c(n)
    integer :: num, lth, ubd, i
    num = n
    lth = 1
    do while (num.gt.1)
      do i = 1, num/2
        ubd = min(2*i*lth,n)
        call merge(c(2*(i-1)*lth+1:(2*i-1)*lth),c((2*i-1)*lth+1:ubd))
      end do
      lth = 2*lth
      num = (num+1)/2
    end do
    return
  end subroutine merge_sort
  subroutine merge(c1,c2)
    implicit none
    integer(8), intent(inout) :: c1(:), c2(:)
    integer(8) :: c(size(c1)+size(c2))
    integer :: i1, i2, n1, n2
    i1 = 1
    i2 = 1
    n1 = size(c1)
    n2 = size(c2)
    do while (i1.le.n1.and.i2.le.n2)
      if (c1(i1).le.c2(i2)) then
        c(i1+i2-1) = c1(i1)
        i1 = i1+1
      else
        c(i1+i2-1) = c2(i2)
        i2 = i2+1
      end if
    end do
    if (i1.le.n1) then
      c(i1+i2-1:n1+n2) = c1(i1:n1)
    else if (i2.le.n2) then
      c(i1+i2-1:n1+n2) = c2(i2:n2)
    end if
    c1 = c(1:n1)
    c2 = c(n1+1:n1+n2)
    return
  end subroutine merge
end program exam_and_wizard